2. Gyakorlat

Maple ismerkedés

Utasítások lezárása ; vs. :
?parancs (help)
hivatkozás korábbi %, %%, %%%
kétféle felsvessz,
unassign,
whattype
aritmetikai mveletek,
értékadás, kiértékelési lánc,
with, restart,
op, nops,
eval, evalf, evalb,
solve, fsolve, isolate,
cat/||,
sum,

alapvet adattípusok: numerikus típusok, string, symbol, algebrai típusok, sorozat, lista, tömb, indexelés, halmaz
függvények definiálása, hívása
diff, int,
plot,
programozási alapismeretek (vezérlési struktúrák)

```
> restart;
> a := 5;
  b := 4;
> c, d := -1, -2;
> a + b;
> e := c - d;
> e := c - d:
> d/b;
> a*c;
> sum(n^2, n = 1..5);
> %, %%, %%%;
> unassign(a);
> unassign('a');
> a;
> a := 5;
> a := 'a';
> a;
> unassign('c', 'd', 'e');
> c, d := -1, -2;
> with(combinat); #package importálása
> ?with
```

```
> `ez egy hosszú nev változó ékezetes és speciális (&@#*?!)
  karakterekkel` := 0;
> restart;
> a := b;
  b := c;
  c := 1;
> a, b, c;
> c := 10;
> a, b, c;
> b := 2;
> a, b, c;
> a := 3;
> a, b, c;
> whattype(a);
> whattype(g);
> whattype(2/3);
> whattype(0.5);
> whattype("Hello!");
> whattype(x+y);
> whattype(x=y);
> whattype([1, 2, 3, 4, 5]);
> type(b, integer);
> S := "Hello!";

> S[1];
> S[-1];
> S[2..4];
> length(S);
> cat(S, " Bye!");
> S||" Bye!";
> cat(g, S);
> a$5;
> restart;
> eq1 := 3*x + 7 = 11;
> nops(eq1);
> op(eq1);
> op(1, eq1), op(2, eq1);
> lhs(eq1), rhs(eq1);
> solve(eq1, x);
> fsolve(eq1, x);
> solve(3*x + 7);
> isolate(eq1, x);
> assign(%);
> x;
  eq1;
  unassign('x');
> egyenlet1 := 3*x - y = -10;
> egyenlet2 := -x + 2*y = -5;
> solve({egyenlet1, egyenlet2}, {x, y});
```

```
> eval(egyenlet1, {x=-5, y=-5});
> evalf(2/3);
> 2/3;
> (sqrt(5) - 1)/2;
> evalf(%); # evaluate float
> pA := 0.5; pB := 0.2; pAB := 0.1;
  evalb(pA*pB = pAB); # evaluate boolean
> pA*pB = pAB;
> evalb(5 in {1, 10, 11, 6, 7});
> 5 in {1, 10, 11, 6, 7};
```

Kombinatorikus valószínség

Klasszikus valószínségi mez

Klasszikus valószínségi meznek nevezzük az olyan esetet, amikor egy kísérlet lehetséges kimenetelei (az elemi események) véges halmazt alkotnak, és a valószínség is egyenlen oszlik meg köztük: n darab lehetséges kimenet létezik, egyenként $\frac{1}{n}$ érték valószínséggel. Ilyen pl. a pénzfeldobás (2 darab $\frac{1}{2}$ valószínség esemény), a kockadobás (6 darab $\frac{1}{6}$ valószínség esemény), vagy a lottó (pl. az ötöslottón közel 44 millió egyenl valószínség számötös kihúzása mint esemény, egyenként nagyon kicsi (~1/44m) valószínséggel).
Ilyen esetekben a különféle összetett (elemi események összességébl álló) események valószínségével kapcsolatban feltehet kérdések lényegében az adott összetett eseményt alkotó ("kedvez") elemi események, valamint az összes lehetséges kimenetel összeszámlálásának feladatát takarják, hiszen a valószínség számszer mértéke ebben a valószínségi mezben ezek aránya ($p = \frac{k}{n}$) - vagy másként fogalmazva: az elemi esemény (mindre egyenl $\frac{1}{n}$) valószínségét csupán a kedvez esetek darabszámával kell szorozni ($p = k \cdot \frac{1}{n}$). Pl. ha az a kérdés, milyen eséllyel dobunk páros számot a kockával, akkor azt kell megszámolni, mennyi az összes lehetséges dobás, és közöttük hány páros van ($p = \frac{3}{6} = \frac{1}{2}$), ha az a kérdés, mi a valószínsége a lottón az egy szelvénnyel elérhet hármas találatnak, azt kell összeszámolni, hányféleképp tehet összesen fogadás, és ezen belül mennyi az adott nyerszámokra tehet 3 találatot ér fogadás. Elbbit könnyebb, utóbbit nehezebb összeszámolni, de mindkét esetben közös, hogy az összes, és azon belül a kedvez esetek darabszámát kell meghatározni.

Kombinatorika

A matematika kombinatorika nev ága lehetségek összeszámlálásával foglalkozik: módszereket, algoritmusokat, stratégiákat foglal magába (jól definiált körülmények mellett) véges sok, de esetenként igen nagy darabszámú eshetség szisztematikus kezelésére és számbavételére. Az alábbiakban áttekintünk néhányat a kombinatorika alapproblémái közül, melyek egyszerbb feladatoknak akár közvetlenül is megoldásai, bonyolultabbak megoldásának megfelel strukturálással moduljai, további módszereknek alapjai lehetnek. Ezek az alapesetek egymással is

összefüggnek; a felosztás inkább a gyakorlati alkalmazhatósághoz igazodik.
A Maple rendszerben a kombinatorikával kapcsolatos parancsok, függvények külön csomagban hívhatók be.

```
> with(combinat);
```

Permutációk

A permutáció elemek egy lehetséges sorbaállítását jelenti. Az egyik legalapvetbb kombinatorikai feladat a permutációk (lehetséges sorbaállítások) számának meghatározása. Két nevezetes változatot különböztetünk meg attól függen, hogy a sorbaállítandó elemek mindannyian különbözek-e, vagy vannak köztük olyan azonosak, melyek egymás közti felcserélése nincs érzékelhet hatással egy adott sorbaállításkor.

Ismétlés nélküli permutáció, a faktoriális fogalma

Ismétlés nélküli permutációnál sorbaállítandó elemek mindannyian különböznek. (A szóban forgó elemek bármik lehetnek, de ahol lehet, az egyszerség kedvéért gyakran számokkal modellezünk.)
Listában megadott elemek egy-egy permutációja (sorbaállítása) maga is egy lista - ezek a Maple *permute* parancsa generálja ezek összességét (mint listák listáját).

```
> elemek := [1, 2, 3];
> permutaciok := permute(elemek);
> nops(permutaciok);
```

A permutációk számát közvetlenül (a permutációk sokaságának gyakran felesleges generálása nélkül) adja meg a *numbperm* függvény:

```
> numbperm(elemek);
```

Nem kötelez, hogy az elemek számok legyenek. Ha azonban nem szeretnénk a szimbólumok használatakor a változók elnevezésére vonatkozó bizonyos konvenciókba ütközni (pl. szóközt, írásjeleket, speciális karaktereket használunk), akkor a Maple rendszerben általában is javasolt a szimbólumokat ` jelek között (AltGr+7) megadni:

```
> gyumolcsok := [`alma`, `körte`, `barack`, `szilva`]:
  permute(gyumolcsok);
  nops(%);
```

A permutációk szisztematikus felsorolásának egyik útja lehet a döntési fa alkalmazása. Az els elágazási szinten felsoroljuk az összes elemet, mivel bármelyikkel kezddhet a felsorolás. A második elágazási szinten minden elsszint csomópontnál választhatunk az azon az ágon éppen megmaradó elemek közül, s így tovább, amíg marad elem. A képzd fa gyökértl levelekig tartó bejárásai az összes permutációt tartalmazzák - a permutációk száma a fa leveleinek számával egyenl. *n* darab sorbaállítandó elem esetén az els szinten n elágazás, a második szinten (n-1) elágazás van, s így tovább az összesen n darab elágazási szinten, ahol a sor 2 majd 1 elágazással (utoljára nincs valódi választás, mivel a végére megmaradó elem már egyértelm) zárul. Az egyes szinteken tapasztalható elágazások számát összeszorozva megkapjuk a levelek, vagyis az összes sorbaállítás, vagyis az összes permutáció darabszámát.
Ilyen esetben tehát a permutációk száma mindig a számok 1-tl n-ig való összeszorzásával adódik. Ezen gyakran használatos, de nagyobb elemszámok mellett csak körülményesen leírható számítás tömör kifejezésére használjuk a faktoriális fogalmát, amely egy felkiáltójellel jelölt egyoperandusú mvelet.
***n* (különböz) elem ismétlés nélküli permutációinak száma:**
$$n(n-1)(n-2)\ldots\cdot 2\cdot 1 = n!$$

A faktoriális viselkedése

Vizsgáljuk meg a faktoriális viselkedését:

```
> 3!=3*2*1;
> 4!=4*3*2*1;
> 10!;
> 100!;
> 1000!;
```

A faktoriális nagyon gyorsan növekszik; gyorsabban, mint az elemi esetek közül leggyorsabb növekedésként ismert mértani (exponenciális) sorozat. Egy id után bármilyen hányadosú mértani sorozatot "lehagy", hiszen míg azok értéke lépésenként ugyanazzal az aránnyal növekszik, itt maga a növekedési arány is lépésrl lépésre n. Az alábbiakban összehasonlítjuk a faktoriális növekedését exponenciális növekedésekkel:

```
> pontok_faktorialis := [seq([i, i!], i=1..10)];
> plot(pontok_faktorialis, style=point, symbolsize=20,
  title=`faktoriális növekedése`, color=red);
> pontok_mertani_2 := [seq([i, 2^i], i=1..10)];
> plot(pontok_mertani_2, style=point, symbolsize=20,
  title=`2-es alapú exponenciális növekedés`, color=
  blue);
> pontok_mertani_5 := [seq([i, 5^i], i=1..10)]: plot
  (pontok_mertani_5, style=point, symbolsize=20, title=
  `5-ös alapú exponenciális növekedés`, color=green);
> plot([pontok_mertani_2, pontok_mertani_5,
  pontok_faktorialis], style=point, symbolsize=20,
  color=[blue, green, red]);
```

Azt látjuk, hogy $n = 10$ értékig a faktoriális a 2-es alapú exponenciális növekedést túlszárnyalja, de az 5-ös alapút még nem. Ez nyilván azért van így, mert idáig a faktoriális fejldésének kezdeti fázisában még 5-nél kisebb tényezk szerepelnek (szemben azzal, hogy az 5-ös alapú exponenciális sorozat a kezdetektl fix 5-ös szorzóval növekszik) - és bár itt már lépésrl-lépésre 5-nél nagyobb (és folyamatosan növ) a faktoriális növekedési aránya, ez idáig még épp nem volt elég a kezdeti hátrány behozásához. Két lépéssel késbb viszont már bekövetkezik a fordulat:

```
> 11! < 5^11;
> 5^12 < 12!;
```

Bár a klasszikus definícióba nem fér bele a faktoriális értelmezése nullára, azonban a fogalom általánosítása szerint ill. a faktoriálisra épül további kombinatorikai formulák széls esetekben is megfelel viselkedéséhez lehet és érdemes 1-nek értelmezni.

```
> 0!;
```

Ismétléses permutáció

Ismétléses a permutáció, ha a sorbaállítandó elemek között vannak ismétldk - vagyis legalább egyik fajta elembl egynél több van.

```
> elemek := [1, 2, 2, 3, 3, 3];
```

A Maple *permute* eljárása ilyenkor is képes generálni a sorbaállításokat.

```
> permute(elemek);
  nops(%);
```

A darabszámot az erre szolgáló közvetlen függvény is helyesen adja:

```
> numbperm(elemek);
```

Tegyük fel, hogy az összesen n darab elem között rendre k_1, k_2, ... ismétld van.

Képzeljük el, hogy átmenetileg az egyformákat is egyedinek tekintjük (pl. sorszámozzuk ket) - ekkor a sorbaállítások száma $n!$ lesz. Ha levesszük a megkülönböztet jelzést, akkor a sorbaállításokban az egyformák hirtelen megkülönböztethetetlenné válnak, vagyis egyformává válik sok olyan permutáció, ami megkülönböztetéssel még különböz volt. Az egyforma elemek egymás közötti sorrendcseréi már nem látszanak különböznek, ráadásul egy egyforma csoport bármelyik önmagán belül kialakított sorrendjéhez bármely másik csoport önmagán belül kialakított sorrendje társulhat, vagyis bármelyik sorbaállításban az egyforma elemek $k_1!$, $k_2!$, ... darab, önmagukon belüli sorbaállításainak lehetségei szabadon kombinálódhatnak - ez pedig az egyformákon belüli permutációk számának összeszorzását jelenti. Vagyis az elemek átmeneti megkülönböztetésével számolt $n!$ értékkel $k_1!\ k_2!\ k_3!$... -szeresen túlszámoltuk a permutációkat, vagyis ennyivel osztani kell - így kapjuk az ismétléses permutációk számát.

n darab elem ismétléses permutációinak száma, ha köztük rendre k_1, k_2, k_3, ... darab ismétld (egyforma) elem van: $\dfrac{n!}{k_1!\cdot k_2!\cdot k_3!\cdot \ldots}$

Pl. az elz esetre:

```
> 6!/(2! * 3!);
```

A Maple beépített közvetlen függvényt is tartalmaz a számításhoz:

```
> multinomial(6, 1, 2, 3);
```

Vegyük észre, hogy az ismétlés nélküli permutáció az ismétléses permutáció olyan speciális esetének is tekinthet, ahol egyelem csoportok szerepelnek. Csupa 1!=1 értékekkel való osztás esetén a formula meghagyja az eredeti $n!$ értékét.

Variációk

A variáció adott (különböz) elemekbl (n darab) képzett valahány (k) elem sorozat (melyben tehát számít a sorrend). *Ismétlés nélküli* variációnak nevezzük azt a széls esetet, amikor minden elem csak egyszer használható fel a sorozatban, *ismétléses* variációnak pedig azt, amikor bármelyik korlátlan számban (a gyakorlatban persze egy k elem sorozathoz csak legfeljebb k-szor) felhasználható.

A megalkotható sorozatok döntési fával modellezhetk, az ismétléses és az ismétlés nélküli eset közötti különbség csak az, hogy az egyre újabb elágazási szinteken csökken-e vagy megmarad az elágazások száma.

Ismétlés nélküli variáció

A döntési fa els szintjén bármelyik elemet választhatjuk, de ha az ismétldés nem megengedett, akkor a másodikon már csak (a megmaradók közül) eggyel kevesebbet stb. Ez azt jelenti, hogy az egyes szinteken az elágazások száma n-tl egyesével visszafelé csökken, vagyis a lehetségek száma egy k-tényezs szorzat: $n\ (n-1)\ (n-2)\ldots(n-k+1)$. Ez az egyesével visszafelé szorozgatás hasonlít a faktoriális képzésére, csak nem megyünk vissza fixen 1-ig, hanem $(n-k+1)$ mint utolsó tényeznél végetér - vagyis az $(n-k)$-tól visszafelé 1-ig tartó szorozgatás (tehát $(n-k)!$) hiányzik belle, ezért az $\dfrac{n!}{(n-k)!}$ értékkel egyenl.

n elem k-ad osztályú ismétlés nélküli variációinak száma: $\dfrac{n!}{(n-k)!}$

Vegyük észre, hogy ismétlés nélküli esetben mindenképpen $k \leq n$, ezen belül pedig $k = n$ esetben éppen az ismétlés nélküli permutáció esetét kapjuk! Ennek a rokonságnak a jele, hogy a Maple a variációkat is a permute eljárással állítja el. Pl. képezzünk 5 elembl 3-elem ismétlés nélküli variációkat:

```
> permute([1, 2, 3, 4, 5], 3); nops(%);
> 5! / (5 - 3)!;
```

Ismétléses variáció

Az egyes elemek korlátlanul ismételhet felhasználhatósága azt eredményezi, hogy egyrészt nincs korlátozás k értékére, másrészt a döntési fa egyes szintjein az elágazások száma nem csökken. Az ismétléses variációk száma tehát egy olyan k-tényezs szorzat, melyben mindenhol n áll, így az ismétléses variációk száma egyszeren n^k.

n elem k-ad osztályú ismétléses variációinak száma: n^k

Maple-ben ismétléses variációkat is eleállíthatunk a permute eljárással, ha elegend ismétldéssel vesszük fel az elemeket a kiinduló listába (pl. 3 elem sorozat esetében legalább 3-szor ismételve):

```
> permute([1, 1, 1, 2, 2, 2, 3, 3, 3, 4, 4, 4, 5, 5, 5],
  3); nops(%);
> 5^3;
```

Kombinációk

Kombinációról beszélünk, ha n megkülönböztetett tulajdonos között osztunk szét k egyforma objektumot. Ismétlés nélkülinek nevezzük a kombinációt, ha egy tulajdonos legfeljebb egy objektumot kaphat, ismétlésesnek, ha egy tulajdonos többet is. (Mindkét esetben maradhatnak "üresen" tulajdonosok, de az összes objektumnak lesz gazdája.) Tipikus példa kombinációra ajándékok szétosztása emberek között.

Ismétlés nélküli kombináció

Bár jelen kategorizálás szerint a kombinációk egyik alesetében írjuk le, ez az eset kiemelt fontosságú. Noha belefér a fenti definícióba, egyszeren úgy is gondolhatunk rá: **egy n elem halmazból (így tehát szükségképpen különböz elemekbl) hányféleképp választhatunk ki k elem részhalmazokat** (tehát a sorrend megkülönböztetése nélkül). (Tehát pl. hányféleképp választhatjuk ki azt a k darab embert, aki kap (csakis egy) ajándékot.)

Pl. nézzük meg, egy 5 elem halmazból hányféleképp választhatunk 3 elem részhalmazokat (öt elem harmadrend ismétlés nélküli kombinációit). A Maple *choose* függvénye (szemben a *permute* függvénnyel) mindig olyan listákat fog csak generálni, melyek sorrendi átrendezéssel (permutálással) nem vihetk át egymásba:

```
> elemek := [1, 2, 3, 4, 5];
> choose(elemek, 3); nops(%);
```

Mint a példából is látszik, ennek egyik következetes módja lehet, ha a kimeneti listákban tartjuk az elemek valamilyen (pl. nagyságuk, vagy az eredeti sorrendjük szerinti) rendezettségét. Ez nem azt jelenti, hogy az eredményben számítana a felsorolási sorrend (pl. $[1, 2, 3]$ és $[3, 1, 2]$ ugyanazt a kiválasztást takarja), hanem hogy épp erre tekintettel ezzel a kritériummal kerülhet el az ilyesfajta ismételt beszámítása ugyanannak a kombinációnak.

Az ismétlés nélküli kombinációk száma valamilyen alkalmas stratégia definiálása nyomán általánosságban is számítható. Egy ilyen lehetséges stratégia: az n elem sorbaállítása után kiválasztjuk a sor elejére kerül k elemet. Az összes elem sorbaállításainak száma $n!$, de ezek között sok ugyanazon adott k elem kiválasztásához vezet: az els k "bejutó" helyen belül és az $n-k$ "kimaradó" helyen belül tetszlegesen permutálhatjuk az elemeket, ugyanaz a k elem választódik ki. Vagyis ugyanazon elemek kiválasztásához az összes sorbaállításon belül $n!\ (n-k)!$ sorbaállítás vezet, eszerint a kiválasztási lehetségek számát úgy kapjuk, ha az összes sorbaállítások számát elosztjuk ezzel az értékkel:

$$\frac{n!}{k!\ (n-k)!}$$

n elem k-ad osztályú ismétlés nélküli kombinációinak száma: $\dfrac{n!}{k!\ (n-k)!} = \dbinom{n}{k}$

Ez egy olyan kiemelt fontosságú formula, hogy külön jelölést is kapott: $\dbinom{n}{k}$ - "n alatt a k".

Ezt mint két paraméteres mennyiséget ún. binomiális együtthatónak is szokás nevezni, melynek számítására közvetlen beépített függvény is van, mely példánkra azonnal megadja a kombinációk darabszámát:

```
> binomial(5,3);
```

Észrevehetjük, hogy az $\dfrac{n!}{k!\ (n-k)!}$ formula megegyezik összesen n darab, kétféle elemet (k ill. $n-k$ darabot) tartalmazó ismétléses permutációk számával. Ez összhangban áll pl. egy olyan stratégiával, melyben az elemeket színek társításával választjuk: mondjuk embereket színes pólók kiosztásával. A sorbaállított emberek számára kiosztott kétféle szín pólók felvételével kialakuló színsorozatok ismétléses permutációkat jelentenek, melyek egyben alkalmasak kiválaszásra is (pl. "kék pólósok lépjenek ki"), tehát az ismétléses permutációk az ismétlés nélküli kombinációkkal párosíthatók - számuk egyenl.
Az ismétlés nélküli kombinációk az ismétlés nélküli variációkból is származtathatók. Mivel az ismétlés nélküli variációk és kombinációk között csupán az a különbség, hogy míg elbbinél megkülönböztetjük a sorrendet, utóbbinál nem, utóbbiak számát megkapjuk, ha a variációk számát ($\dfrac{n!}{(n-k)!}$) a kiválasztottakon belül elállítható sorrendek számával ($k!$) osztjuk, ami ugyancsak a fenti képletre vezet.
Vegyük észre azt a szimmetriát, miszerint egy n elem alaphalmaz esetén k ill. $n-k$ elem kiválasztási lehetségeinek száma egyenl. A formulába helyettesítés is ezt mutatja, és könnyen érthet is, tekintve, hogy k elem kiválasztása közvetve a megmaradó $n-k$ elem kiválasztását is jelenti.

Binomiális együtthatók

Az ismétlés nélküli kombinációk számának kiszámítására szolgáló, két paramétert tartalmazó $\dbinom{n}{k}$ formulát szokás binomiális együtthatónak is nevezni, mivel (az általa leírt kombinatorikai tartalommal összhangban) ezek az együtthatók jelennek meg egy kéttagú összeg n-edik hatványának kifejtésében, és ezek alkotják az ún. *Pascal-háromszög*et is.

Kéttagú kifejezés második hatványa közismerten:

```
> expand((a+b)^2);
```

Ennek általánosításaként két tag egyre magasabb kitevj hatványainak kifejtése (az ún.

binomiális tétel szerint):

```
> for n from 0 to 10 do
    expand((a+b)^n);
  end do;
```

A tagok tényezinek hatványkitevi jól követhet mintázatot mutatnak (balról-jobbra haladva az a kitevje csökken, a b kitevje n). Ha ezektl eltekintve a csak a mellettük álló számszer együtthatókat figyeljük, azok a következ mintázatot adják (két Maple-megoldás szerint is kiírva):

```
> for n from 0 to 10 do
    row:="":
    for k from 0 to n do
      row:=cat(row, cat(binomial(n,k), " ")):
    end do:
    print(cat(" ", row));
  end do:
```

```
> plottools:-rotate(plots:-textplot([seq(seq([i-j, j,
  binomial(i, j)], j = 0 .. i), i = 0 .. 10)], axes =
  none), -3*Pi*(1/4));
```

Ez az ún. *Pascal-háromszög*, melynek elemei épp a binomiális együtthatók, melyek érdekes összefüggései mutatkoznak meg ebbl az elrendezésbl:

- Az n. sor k. eleme épp $\binom{n}{k}$ (a sorok és azon belül az elemek számozását is nullától kezdve).
  Ez azért van így összhangban a kéttagú összeg hatványainak kifejtésével, mert pl. $n=5$ és $k=3$ értékekhez a kifejtésben az a $10\,a^2b^3$ tag tartozik, mely épp azokból a kombinációkból áll össze, amikor az $(a+b)^5=(a+b)\cdot(a+b)\cdot(a+b)\cdot(a+b)\cdot(a+b)$ szorzat felbontásakor az 5 darab kéttagú tényez közül éppen 2-bl választjuk az a-t és 3-ból a b-t. Mivel ez kombinatorikailag $\binom{5}{3}=10$ féle módon következhet be, a kifejtés során keletkez ilyen tagok összesen ennyien lesznek, melyekbl összevonás után áll össze a $10\,a^2b^3$ tag. Ez mondható el az összes többire is, ami maga **a binomiális tétel, melynek precíz alakja:**
  $$\boldsymbol{(a+b)^n=\binom{n}{0}a^n+\binom{n}{1}a^{n-1}b+\binom{n}{2}a^{n-2}b^2+\ldots+\binom{n}{n-1}ab^{n-1}+\binom{n}{n}b^n}$$
  $$\boldsymbol{=\sum_{k=0}^{n}\binom{n}{k}a^{n-k}b^k}$$
- Az alakzat szimmetriája összhangban van a mögötte álló kombinatorikai feladat szimmetriájával (a kiválasztással komplementer értelemben a nem-kiválasztottak is kiválasztódnak).
- A Pascal-háromszög peremén 1-esek állnak, mivel $\binom{n}{0}=\binom{n}{n}=1$
- A Pascal-háromszög belsejében álló (nem 1 érték) minden elem az ábrázolt elrendezés szerint a felette (jobbra és balra) álló elemek összege. Formálisan ez a binomiális együtthatók között érvényes $\binom{n+1}{k+1}=\binom{n}{k}+\binom{n}{k+1}$ azonosság.
  Ennek szemléletesen érthet kombinatorikai tartalma az, hogy eggyel több $(n+1)$ elembl eggyel több $(k+1)$ elem kiválasztásának problémája felépíthet annak a két

esetnek az egyesítésébl, hogy 1.) a kiválasztásba garantáltan bekerül az új elem és ekkor a maradék k darabot az eredeti n közül választjuk ($\binom{n}{k}$ lehetség), vagy 2.) a kiválasztásba garantáltan *nem* kerül be az új elem és így *mindet* az eredetiek közül választjuk ($\binom{n}{k+1}$ *lehetőség*).

Ez formálisan is levezethet. Az egyenlség két oldalát a definíció szerint kifejtve:

A bal oldal:

$$\binom{n+1}{k+1} = \frac{(n+1)!}{(k+1)!\cdot((n+1)-(k+1))!} = \frac{(n+1)!}{(k+1)!\cdot(n-k)!}$$

A jobb oldalon a kifejtés után a tört tagok megfelel átalakításával egyenként próbáljuk megközelíteni a fenti eredmény nevezjét, majd a közös nevezj törtek összeadásával belátni, hogy az összeg számlálója (és így a két eredeti kifejezés is) megegyezik a két oldalon:

$$\binom{n}{k} + \binom{n}{k+1} =$$

$$= \frac{n!}{k!\cdot(n-k)!} + \frac{n!}{(k+1)!\cdot(n-(k+1))!} =$$

$$= \frac{n!}{k!\cdot(n-k)!} + \frac{n!}{(k+1)!\cdot(n-k-1)!} =$$

$$= \frac{n!(k+1)}{(k+1)!\cdot(n-k)!} + \frac{n!}{(k+1)!\cdot\frac{(n-k)!}{n-k}} =$$

$$= \frac{n!(k+1)}{(k+1)!\cdot(n-k)!} + \frac{n!(n-k)}{(k+1)!\cdot(n-k)!} =$$

$$= \frac{n!(k+1+n-k)}{(k+1)!\cdot(n-k)!} =$$

$$= \frac{n!(n+1)}{(k+1)!\cdot(n-k)!} =$$

$$= \frac{(n+1)!}{(k+1)!\cdot(n-k)!}$$

- A Pascal-háromszög n. sorában az elemek összege éppen 2^n - pl. az 5. sorra:
  $1+5+10+10+5+1=32=2^5$
  Ez könnyen indokolható azzal, hogy az n. sor binomiális együtthatóinak összege éppen az $(1+1)^n$ kifejtésével állítható el (ekkor a kifejtés minden egyes tagjában lév 1 alapú hatványok értéke 1).

Ismétléses kombináció

Ismétléses kombináció esetén úgy osztunk szét *egyforma* objektumokat tulajdonosok között, hogy egy tulajdonos többet is kaphat.

Pl. tekintsük azt a feladatot, hogy 5 ember között osztunk szét 3 egyforma tárgyat úgy, hogy egyvalaki többet is kaphat. Egy-egy lehetséges kiosztást modellezhetünk pl. úgy, hogy 3-elem listákat készítünk a tulajdonosokról, akik ajándékot kapnak. Ehhez a tulajdonosokból listát (sokaságot) készítünk, legalább annyiszor megismételve a listában mindenkit, ahány ajándék van (hogy modellezhet legyen az is, hogy széls esetben minden ajándékot egy ember kap), majd ebbl választunk ki megfelel elemszámú (ahány szétosztandó ajándék van) rész-sokaságokat.

```
> tulajdonosok := [1, 1, 1, 2, 2, 2, 3, 3, 3, 4, 4, 4, 5, 5, 5];
> choose(tulajdonosok, 3); nops(%);
```

Látjuk, hogy a Maple szisztematikusan felsorolja a lehetségeket a széls esetek között (mindhárom ajándékot az els ember kapja, kettt kap az els és egyet a második, ... stb. ... mindhármat az utolsó kapja). A felsorolt esetek között az segíti az ismétldések elkerülését, hogy az egyes eseteknél emelked sorszámok szerint sorolja a tulajdonosokat. 35 eset lehetséges.

Hogy lehet kombinatorikai formulát adni a lehetségek számára? Képzeljük el, hogy egy sorban elhelyezünk (hézagosan) $n-1$ darab (az emberek számánál eggyel kevesebb) elválasztó-jelet (pl. "|"). Ez a széleken kialakuló helyekkel együtt éppen n darab tartomány - minden embernek egy. Az ajándékok kiosztását úgy modellezzük, hogy k darab jelet (pl. "+") szórunk el a sorban az elválasztó-jelek között. Két elválasztójel közé és a szélekre is juthat bármennyi (0-tól k-ig): akár minden jel kerülhet egy tartományba, de maradhatnak üresek is. Ezek szerint minden egyes kiosztási lehetség modellezhet egy olyan jelsorozattal, melyet $n-1$ darab "|" szeparátor és k darab "+" jel alkot. A kiosztások száma annyi, ahányféleképp az összesen $n+k-1$ elem jelsorozat összeállhat. Ez akár ismétléses permutációval, akár ismétlés nélküli kombinációval gondolkodva $\binom{n+k-1}{k}$.

Pl. esetünkben:

```
> binomial(5 + 3 - 1, 3);
```

Kidolgozott példafeladatok

1. Kártya keverése

a) Hányféle keverése létezik a "magyar kártya" paklinak?

b) Mekkora a valószínsége, hogy a kevert pakli alján épp a "zöld alsó", tetején a "piros ász" lesz?

Megoldás

```
> restart; with(combinat):
```

a)

A kártyák halmaza:

```
> Deque := [seq(seq(cat(i, j), j in ["7", "8", "9", "10", "A", "F", "K", "Á"]), i in ["Z", "P", "T", "M"])];
> n := nops(Deque); # 32-t várunk
```

A lehetséges keverések számát az n kártya permutációinak száma adja (minden kártyából 1 db van a pakliban és a sorrend számít):

```
> N := n!; # factorial(n) is mködik
```

Eredmény konvertálása normálalakba: Jobb klikk az eredményre > Numeric formatting > Scientific

Állítsuk át a tizedesjegyek számát 3-ra.

Ugyanezt az eredményt kapjuk, ha közvetlenül a Deque permutációit számoljuk össze.

```
> numbperm(Deque);
```

b)

Az eseménytér: $\Omega = \{\textit{magyar kártya pakli keverései}\}$.

Ennek számosságát épp az a) pontban határoztuk meg.

```
> nOmega := N;
```

Jelölje A azt az eseményt, hogy a kevert pakli alján épp a "zöld alsó", tetején a "piros ász" lesz. A $P(A)$ valószínséget kell meghatároznunk. Mivel bármely kártyasorrend azonos valószínséggel áll el, ezért a valószínségi mez klasszikus, így a $\frac{\text{kedvező eset}}{\text{összes eset}}$ képlettel számolhatunk.

Kedvez esetek:
Ha lerögzítjük a pakli alján és tetején lév lapot, akkor csak a maradék 30 lapot kell sorba rendezni. Ezek halmaza:

```
> Deque2 := remove(x -> x in {"PÁ", "ZA"}, Deque);
> k := nops(Deque2); # most 30-at várunk
```

A kedvez esetek száma a 30 lap permutációinak száma:

```
> nA := k!;
```

Eredmény konvertálása normálalakba: Jobb klikk az eredményre > Numeric formatting > Scientific, majd állítsuk át a tizedesjegyek számát 3-ra.

Ugyanezt kapjuk, ha közvetlenül a Deque2 permutációit számoljuk össze.

```
> numbperm(Deque2);
```

A keresett valószínség:

```
> PA := nA/nOmega; evalf(%); # az evalf paranccsal
  lebegpontossá alakítjuk az elzleg kiszámolt
  valószínséget
```

Nagyjából 1000 keverésbl 1-szer következik be az A esemény.

2. Kettes találat a lottón

a) Egy adott héten hányféle szelvény-kitöltés mellett lehet kettes találatunk a lottón?
b) Mennyi esélyünk van a kettes találatra?

Megoldás

```
> restart; with(combinat):
```

a)

Képzeljük el egy pillanatra, hogy tudjuk az adott héten kihúzott 5 számot! Az általunk kitöltött szelvénnyel akkor érünk el kettes találatot, ha a kihúzott 5 számból pontosan kettt, a maradék 85 számból pedig pontosan hármat jelölünk be.

Az 5 nyerszámból kettt kiválasztani $\binom{5}{2}$-féleképpen lehet, hiszen a kiválasztás sorrendje nem számít és egy számot csak egyszer választhatunk (ismétlés nélküli kombináció).

Hasonlóan, a 85 nem-nyer számból hármat kijelölni $\binom{85}{3}$-féleképpen lehet.

A fent leírt két kiválasztás egymástól függetlenül megy végbe, így a szorzás szabály alapján a szorzatuk adja a keresett esetek számát:

```
> K := binomial(5,2)*binomial(85,3);
```

b)

Az eseménytér: $\Omega = \{\text{lottószelvény különböző kitöltései}\}$.

Jelölje A azt az eseményt, hogy kettes találatunk lesz a lottón. A $P(A)$ valószínséget kell meghatároznunk. Mivel véletlenszeren töltjük ki a szelvényt, bármely kitöltés azonos valószínség, így a valószínségi mez klasszikus, a $\frac{kedvező\ eset}{összes\ eset}$ képlettel számolhatunk.

Az összes esetek száma, azaz az eseménytér számossága egy ismétlés nélküli kombinációval írható le, mivel 5 (különböz) számot kell kiválasztani a 90-bl.

```
> nOmega := binomial(90,5);
```

A kedvez esetek számát az a) pontban határoztuk meg.

```
> nA := K; #kedvez esetek száma
```

A keresett valószínség:

```
> PA := nA/nOmega; evalf(%);
```

Tehát nagyjából 2.2% eséllyel lesz kettes találatunk.

3. Új rendszámtáblák

A Magyarországon nemrég bevezetett újfajta rendszámtáblák 4 betbl (*A-Z*) és 3 számjegybl (*0-9*) állnak. Hány új gépjárm regisztrálható, mieltt ismét új rendszám-formátumot kellene bevezetni?

Megoldás

```
> restart; with(combinat):
```

A kérdés a lehetséges új típusú rendszámok száma. Összesen $k=7$ helyre kell szimbólumot választani. Az els 4 hely mindegyikére 26 bet közül, míg az utolsó 3 helyre 10 számjegy közül választhatunk. A kiválasztások egymástól függetlenek, így a szorzás-szabályt alkalmazhatjuk az esetek leszámlálására.

```
> k := 7;
  k1 := 4; # betk száma
  k2 := 3; # számjegyek száma
  n1 := 26; # betk száma
  n2 := 10; # számjegyek száma
```

$$k := 7$$
$$k1 := 4$$
$$k2 := 3$$
$$n1 := 26$$
$$n2 := 10 \quad \textbf{(2.3.3.1.1)}$$

```
> N := n1^k1*n2^k2;
```

$$N := 456976000 \quad \textbf{(2.3.3.1.2)}$$

Megjegyzés: úgy is gondolhatunk a problémára, hogy két ismétléses variáció szorzatát vesszük, melyek a 4 betbl, ill. a 3 számjegybl álló szimbólum-sorozatok számát adják meg. Azért ismétléses variációról van szó, mert számít a szimbólumok sorrendje és egy szimbólum többször is elfordulhat. A tanult n^k képlet alapján így az összes rendszámtáblák száma: $N = n_1^{k_1} \cdot n_2^{k_2} = 26^4 \cdot 10^3$.

4. Számok alkotása számjegyekbl

a) Hány különböz háromjegy szám alkotható az 1, 2, 3, 4, 5 számjegyekbl, ha egy számjegy csak egyszer használható fel?

b) Mekkora valószínséggel lesz a szám páros?

Megoldás

```
> restart; with(combinat):
```

a)

A számjegyek listája:

```
> S := [1, 2, 3, 4, 5];
```

Három (különböz) számjegyet kell választanunk az 5-bl, és a sorrend is számít, ami egy ismétlés nélküli variáció. Mivel egyik számjegy sem 0, ezért minden esetben érvényes háromjegy számot kapunk.

```
> n := nops(S); # felhasználható számejegyek száma
> k := 3; # kiválasztandó számjegyek száma
```

A megoldás:

```
> N := n!/(n-k)!; # vagy N := numbperm(S, 3);
```

A *permute* paranccsal tudjuk felsorolni az összes elkészíthet 3 jegy számot.

```
> L := permute(S, 3);
```

Ellenrizhetjük, hogy jól számoltunk-e:

```
> evalb(nops(L)=N);
```

b)

Az eseménytér: $\Omega = \{1, 2, 3, 4, 5\ \textit{számjegyekből képezhető}\ 3\ \textit{jegyű számok}\}$.
Ennek számosságát épp az a) feladatrészben határoztuk meg.

```
> nOmega := N;
```

Jelölje A azt az eseményt, hogy a kapott 3 jegy szám páros. A $P(A)$ valószínséget kell meghatároznunk. Mivel minden elkészíthet 3 jegy szám azonos valószínséggel adódik, ezért a $\dfrac{\textit{kedvező eset}}{\textit{összes eset}}$ képlettel számolhatunk.

Kedvez esetek száma:

Páros számot úgy kaphatunk, ha az utolsó számjegy páros, azaz 2 vagy 4. A 3. helyre így 2 számjegy közül választhatunk.
A 2. helyre már nem választhatjuk az elbbi számjegyet, de bármely másikat igen, így 4 választási lehetségünk van.
Az 1. helyre a korábbiakat már nem választhatjuk, így 3 választási lehetségünk marad. A kedvez esetek száma a szorzás szabály alapján tehát:

```
> nA := 2*4*3;
```

A kedvez eseteket fel is sorolhatjuk a *select* eljárás segítségével:

```
> L2 := select(i -> i[3] in {2,4}, L);
```

Ellenrizhetjük, hogy jól számoltunk-e:

```
> evalb(nops(L2)=nA);
```

A keresett valószínség:

```
> PA := nA/nOmega; evalf(%);
```

Tehát 40% eséllyel kapunk páros számot. (Ez úgy is látszik, ha csak az utolsó számjegyet vizsgáljuk: az 5 lehetséges számjegy mindegyike azonos valószínséggel kerülhet az utolsó helyre (egyik sem kitüntetett), és közülük 2 jó, 3 rossz, így $\frac{2}{5}$ a kérdéses valószínség.)

5. Szó kiolvasása bettáblázatból

Hányféleképp olvasható ki egy 6x5-ös táblázatból a MATEMATIKA szó?

Megoldás

```
> restart; with(combinat):
```

A táblázat:

```
> T := Matrix(6, 5, (i,j)->"MATEMATIKA"[i+j-1]);
```

A kérdés arra vonatkozik, hogy a bal fels sarokból hányféleképpen lehet eljutni a jobb alsó sarokba úgy, hogy csak jobbra (J) és lefelé (L) lépéseket alkalmazunk. Könnyen látható, hogy minden ilyen lépéssorozat a MATEMATIKA szó egy kiolvasását adja, mivel az "M" bettl indulunk és bármely bettl jobbra vagy lefelé lépve a szó következ betjét kapjuk.

A MATEMATIKA szó egy kiolvasása tehát megfelel egy olyan lépéssorozatnak, melyben csak jobbra és lefelé lépünk és a bal fels sarokból a jobb alsó sarokba jutunk. Például *JJLLLJJLL*. Vegyük észre, hogy minden ilyen lépéssorozatnak 9 lépésbl kell állnia, melyek között pontosan 4 darab J és 5 darab L lépés van (mindenképp 4-et kell jobbra lépnünk ahhoz, hogy az elsbl az utolsó oszlopba jussunk és 5-öt kell lefelé lépnünk ahhoz, hogy az elsbl az utolsó sorba jussunk).

Fordítva, minden 4 J és 5 L betbl képzett sorozat egy olyan lépéssorozatot kódol, mellyel a bal fels sarokból indulva a jobb alsó sarokba jutunk.

A feladat megoldása tehát az ilyen lépéssorozatok száma. Rögzített példányszámú szimbólumok összes lehetséges sorrendjét kell megszámolnunk, ami egy ismétléses permutáció.

```
> j, l := 4, 5; n := j+l;
```

A megoldás:

```
> N := n!/(j!*l!); # N := numbperm(["J"$4, "L"$5]); is jó
```

Tehát 126-féleképpen olvasható ki a MATEMATIKA szó a táblázatból.

A *permute* paranccsal felsorolhatjuk az összes megfelel lépéssorozatot.

```
> L := permute(["J", "J", "J", "J", "L", "L", "L", "L",
  "L"]);
```

Ellenrizhetjük, hogy jól számoltunk-e:

```
> evalb(nops(L)=N);
```

6. Kéttagú összeg hatványának kifejtése

$(a+b)^{12} = ?$

Megoldás

```
> restart; with(combinat):
```

A binomiális tételt alkalmazzuk: $(a+b)^n = \sum_{k=0}^{n} \binom{n}{k} \cdot a^{n-k} b^k$.

Megjegyzés: fenti képlet a következképpen bizonyítható: ha az $(a+b)^n$ n-tényezs szorzatban elvégezzük a szorzásokat (felbontjuk a zárójeleket, minden tagot minden taggal szorzunk), akkor az eredmény minden tagja úgy adódik, hogy minden $(a+b)$ tényezbl kiválasztjuk a-t vagy b-t és ezeket összeszorozzuk. Ha k-szor választunk b-t, akkor értelemszeren $(n-k)$-szor választunk a-t, tehát az eredmény $a^{n-k} b^k$ alakú tagokból áll. Egy ilyen tag összesen $\binom{n}{k}$-szor adódik, mivel ennyiféleképpen lehet k darab b-t kiválasztani az n tényezbl.

```
> n := 12;
> sum(binomial(n, k)*a^(n-k)*b^k, k = 0..n);
```

Az *expand* paranccsal is elvégezhet a zárójelek felbontása:

```
> expand((a+b)^n);
```

Láthatjuk, hogy a két kifejezés megegyezik. Az együtthatókat akár Pascal-háromszögbl is kinyerhetjük, ha felépítjük a 12. szintig.

7. Kockacukrok szétosztása

10 ember összesen 8 kockacukrot kap a kávéjához - hányféleképp osztozhatnak meg rajtuk?

Megoldás

```
> restart; with(combinat):
```

Minden egyes kockacukornál el kell dönteni, hogy melyik ember kapja. Így összesen 8 embert választunk a 10-bl. A sorrend nem számít (a kockacukrok egyformák), viszont egy ember több kockacukrot is kaphat (akár mind a 8-at), így ismétléses kombinációról van szó.

```
> n := 10; # ennyi emberbl választunk
> k := 8; # ennyi embert választunk ki
```

A megoldás:

```
> N := binomial(n+k-1,k);
```

A kockacukrok összes lehetséges elosztása a *choose* paranccsal íratható ki. Ehhez elbb el kell készíteni az embereknek egy olyan listáját, ahol minden ember 8-szori ismétléssel szerepel.

```
> emberek := [seq(i$k, i = 1..n)];
> L := choose(emberek, k): L[1..30]; # az output nagyon
  hosszú lenne, ezért csak az els néhány elemét íratjuk
  ki!
```

Ellenrizhetjük, hogy jól számoltunk-e:

```
> evalb(nops(L)=N);
```

8. Pénzfeldobás-sorozat

a) Egy érme tízszeri feldobásakor milyen valószínséggel lesz pontosan 3 fej az eredményben?
b) Milyen valószínséggel lesz ennél több?
c) Milyen valószínséggel lesz éppen 5 írás?

Megoldás

```
> restart; with(combinat):
```

a)

Az eseménytér: $\Omega = \{10 \text{ } \textit{hosszú Fej} - \textit{Írás sorozatok}\}$

A kísérlethez tartozó valószínségi mez klasszikus, mivel minden 10 hosszú Fej-Írás sorozat ugyanakkora valószínséggel adódik (az érme szabályos). A keresett valószínség számolható a $\frac{\textit{kedvező eset}}{\textit{összes eset}}$ képlettel.

```
> n := 10;
```

A 10 dobás egymástól független és mindegyiknél két kimenetel lehetséges, F vagy I. Így az összes esetek száma:

```
> nOmega := 2^n;
```

A kedvez esetek számát úgy kapjuk, ha megszámoljuk az olyan 10 hosszú F-I sorozatokat, melyekben pontosa 3 darab F szerepel. A 3 F helyét $\binom{10}{3}$-féleképpen választhatjuk ki (ismétlés nélküli kombináció).

```
> k := 3;
> nA := binomial(n,k);
```

Megjegyzés: ismétléses permutációval is számolhattunk volna, ahol 3 Fejet és 7 Írást kell sorba rendezni.

A keresett valószínség:

```
> PA := nA/nOmega; evalf(%);
```

b)

Jelölje B azt az eseményt, hogy 3-nál több fejet dobunk. Érdemes az ellentét esemény valószínségét kiszámolni: $P(\overline{B}) = P(\mathit{legfeljebb}\ 3\ \mathit{Fejet\ dobunk}) = 1 - P(B)$.

Az összeadás $-$ szabály alapján,
$P(\overline{B}) =$
$P(\mathit{nem\ dobunk\ Fejet}) + P(1\ \mathit{Fejet\ dobunk}) + P(2\ \mathit{Fejet\ dobunk}) + P(3\ \mathit{Fejet\ dobunk})$
Az összeadandókat az a) részhez hasonlóan számítjuk ki.

```
> PBc := binomial(10,0)/nOmega+binomial(10,1)/nOmega+
  binomial(10,2)/nOmega+binomial(10,3)/nOmega; evalf(%);
```

A megoldás:

```
> PB := 1-PBc; evalf(%);
```

c)

Jelölje C azt az eseményt, hogy pontosan 5 Írást dobunk. Az a) részhez hasonlóan okoskodhatunk, csak most Fej helyett az Írások számát tekintjük.

```
> nC := binomial(10,5);
  PC := nC/nOmega; evalf(%);
```

9. Születésnapok egyezése

Egy embercsoportban (baráti társaság, iskolai osztály stb.) elfordulhatnak azonos napon születettek. Lehet, hogy ki-ki találkozott is már olyan esettel, hogy csodálkozva felfedezték, a csoportban vannak azonos születésnaposok. Megvizsgálhatjuk valószínségelméletileg, milyen körülmények között mennyire számít különlegesnek ez a helyzet.

A helyzet modellezhet pl. egy olyan elképzelt kísérleti konstrukcióval, hogy az utcáról véletlenszeren behívott emberek között vizsgáljuk a születésnapok egyezését. (A felesleges bonyolítást elkerülend figyelmen kívül hagyjuk a szöknapokat; 365 napos évekkel, és ugyanennyi lehetséges születésnappal számolunk.)

A feladatot a következképp fogalmazzuk meg: **mennyi az esélye, hogy n db függetlenül választott személy között lesz legalább kett, aki az év azonos napján (hónap, nap) született?**

A feladatot bármilyen pozitív n-re (legalább egy, de véges sok emberre) értelmezhetjük, és részletesebb számítások nélkül is rögtön tehetünk néhány megállapítást. Nyilván kettnél kevesebb ember esetén nincs értelme feltenni a kérdést, és az is könnyen érthet, hogy 366 vagy több ember esetén már teljesen biztosan lesznek azonos születésnaposok, hiszen ha 365 embernek kötelezen különböz születésnapokat is osztunk, a 366. már csak foglalt napra kerülhet, egyezést eredményezve (skatulya-elv).

Az általános megoldás eltt nézzünk speciális eseteket:

Speciális esetek

Mekkora eséllyel egyezik meg két személy születésnapja?

Két ember esetén mindkettnek egymástól függetlenül 365 választási lehetség van a születésnapra, így az összes esetek száma:

```
> ket_ember_osszes := 365 * 365;
```

Akkor egyezik meg a születésnapjuk, ha mindketten az év els napján, vagy mindketten a második napján stb. születtek - ez összesen 365 lehetség.

```
> ket_ember_van_egyezes := 365;
```

Ezek nyomán az azonos születésnap esélye két ember esetén:

```
> P[2] := ket_ember_van_egyezes / ket_ember_osszes;
  evalf(%);
```

Ez az eredmény összhangban áll egy másik gondolatmenettel megkaphatóval (st, meg is sejteti azt). Onnan is kiindulhatunk ugyanis, hogy az els ember eleve adott valamilyen születésnappal, és csak a másodiktól kezdünk figyelni: a már kiválasztott ember születésnapjával való egyezésre az év összes lehetséges napja közül 1/365 az esély.

Mekkora eséllyel lesznek három személy között azonos születésnaposok?

Vegyük észre, hogy nem az a követelmény, hogy mindhármuk születésnapja megegyezzen, hanem hogy legyen köztük legalább kett!

Képzelhetjük azt, hogy egymás után hívjuk be az embereket az utcáról. Mindegyiküknek egymástól függetlenül lehet születésnapja az év 365 napjának bármelyikén:

```
> harom_ember_osszes := 365 ^ 3;
```

Össze kellene számolni, hogy ezek közül hány esetben lesz legalább két azonos születésnap.
Az els behívásával rögzítünk egy születésnapot (365 lehetség).

A második behívásakor két eset lehetséges: megegyezik a születésnapja az els emberével (1 lehetség), vagy sem (364 lehetség).

Ha megegyezik, akkor a harmadik személy már tetszleges születésnappal érkezhetne (365 lehetség), az azonos születésnap megvan. Vegyük viszont észre, hogy az így megszámlált esetek között az is szerepel, amikor mindhármuk születésnapja azonos! Ha kizárólag az els két személy születésnapjának egyezéseit szeretnénk számlálni, akkor a harmadik személy lehetségei közül ki kell vennünk az els két személy születésnapját, így nála csak 364 lehetséggel számolhatunk.

```
> pont_elso_masodik_azonos := 365 * 1 * 364;
```

Ha nem egyezik meg a második személy születésnapja az elsével (365*364 lehetség), akkor egyelre nincs azonos születésnap, a harmadik ember behívásakor viszont az egyezés rögtön kétféleképp is elállhat, mivel a két bent lév közül bármelyikkel születhetett azonos napon (1-1 féle módon):

```
> pont_elso_harmadik_azonos := 365 * 364 * 1;
  pont_masodik_harmadik_azonos := 365 * 364 * 1;
```

Nem zárható ki az sem, hogy mindhármuk születésnapja megegyezik (mint azt egy fentebbi esetnél említettük, és a következetesség kedvéért ott ki is szrtük):

```
> elso_masodik_harmadik_azonos := 365 * 1 * 1;
```

Legalább két személy születésnapjának egyezése a fenti esetek összességébl áll össze:

```
> legalabb_ket_azonos := pont_elso_masodik_azonos +
  pont_elso_harmadik_azonos +
  pont_masodik_harmadik_azonos +
  elso_masodik_harmadik_azonos;
```

Ebbl a valószínség:

```
> P[3] := legalabb_ket_azonos / harom_ember_osszes;
  evalf(%);
```

Láttuk tehát, hogy a három emberre vonatkozó esetnél már több kombinációs lehetséget kellett végigszámolni. Ez a bonyolultság újabb személyek belépésével drasztikusan n - pl. gondoljunk arra, hogy több személy esetén több születésnap körül alakulhatnak ki változó létszámú csoportok. Ezek egyre körülményesebb végigvitele helyett érdemes stratégiát váltani.

Az általános megoldás

Példánk jól szemlélteti azt a helyzetet, mikor a feladatban kérdezett esemény valószínsége helyett érdemes megpróbálni a fordított (komplementer) esemény tárgyalását, mert közvetve annak valószínsége is választ ad az eredeti kérdésre, hiszen ha annak P valószínségét tudnánk, 1 mínusz annyi lesz az eredeti kérdésre a válasz. Márpedig olykor a komplementer esemény elemzése könnyebb.

Változtassunk stratégiát: számláljuk a komplementer eseményeket! A "létezik legalább két ember egyez születésnappal" feltétel ellentéte, hogy "nem létezik legalább két ember egyez születésnappal", vagyis "minden ember más napon született". Látni fogjuk, hogy erre a feltételre sokkal könnyebb a megfelel esetek számlálása.

Két személy esetében ez azt jelenti, hogy az els tetszlegesen választott születésnapja (365 lehetség) mellett a másodiknak már csak az ettl külöbözeket választhatjuk (364):

```
> ket_ember_nincs_egyezes := 365 * 364;
```

Ebbl az eredeti feltételeknek megfelel darabszám megegyezik az ott közvetlenül számítottal:

```
> ket_ember_van_egyezes := ket_ember_osszes -
  ket_ember_nincs_egyezes;
```

Három vagy több ember esetében sem nehezebb a dolog; egyszeren annyi, hogy az újabb és újabb belépk számára mindig eggyel kevesebb nap áll rendelkezésre az addigiakkal való egyezés garantált elkerüléséhez.

```
> harom_ember_nincs_egyezes := 365 * 364 * 363;
```

```
> legalabb_ket_azonos := harom_ember_osszes -
  harom_ember_nincs_egyezes;
```

Ezt próbáljuk általánosítani!

```
> emberek_db := 3; napok_db := 365;
> osszes_lehetoseg := napok_db ^ emberek_db;
> nincs_egyezes := napok_db! / (napok_db - emberek_db)!;
> van_egyezes := osszes_lehetoseg - nincs_egyezes;
> van_egyezes_valoszinuseg := van_egyezes /
  osszes_lehetoseg; evalf(%);
```

Az általánosítás sikerült - a fenti mveletsorban az emberek számát változtatva, majd újra futtatva a parancsokat kiszámítható az egyez születésnap létezésének valószínsége. Még igényesebb megoldáshoz jutunk, ha a kiszámító algoritmust egy - az emberek számával paraméterezhet - procedúrába foglaljuk:

```
> van_egyezes_valoszinuseg := proc(emberek_db)
>   local napok_db, van_egyezes_valoszinuseg;
>   napok_db := 365;
>   van_egyezes_valoszinuseg := 1 - (napok_db! / (napok_db
  - emberek_db)!) / (napok_db ^ emberek_db);
>   return van_egyezes_valoszinuseg;
> end proc;
```

Ez a procedúra közvetlenül megválaszolja a kérdést:

```
> van_egyezes_valoszinuseg(3); evalf(%);
```

```
> van_egyezes_valoszinuseg(10): evalf(%);
  van_egyezes_valoszinuseg(20): evalf(%);
  van_egyezes_valoszinuseg(30): evalf(%);
  van_egyezes_valoszinuseg(40): evalf(%);
```

Próbálkozással megtalálható, milyen létszámnál lép át a valószínség egy adott értéket, pl. 50%-ot (23 személynél):

```
> evalf(van_egyezes_valoszinuseg(22)); evalf
  (van_egyezes_valoszinuseg(23));
```

A procedúra ciklikus hívásával könnyen végigszámolhatók a valószínségek az összes értelmezhet csoportlétszámra:

```
> pontok := [seq([i, van_egyezes_valoszinuseg(i)], i = 2.
  .365)]:
> plot(pontok, style = point);
```

Az ábráról leolvasható, hogy a létszám növekedésével eleinte rohamosan n az egyez születésnapok elfordulásának valószínsége, majd telítdést tapasztalunk (hiszen nem lépheti túl az 1-et). (Kb. 130-tól annyira megközelíti a valószínség az 1-et, hogy a számításoknál kerekítési hibák lépnek fel, és nem is ábrázolja tovább.)

```
> van_egyezes_valoszinuseg(40): evalf(%);
  van_egyezes_valoszinuseg(50): evalf(%);
  van_egyezes_valoszinuseg(60): evalf(%);
```

Látható, hogy 40 fnél már nagyon valószín (~90%), 50 fnél szinte biztos (~97%) az azonos születésnapok elfordulása.

10. Klub elnöksége

Egy 25 fős klub 3 tagú vezetőséget –elnököt, titkárt és jegyzőt– választ. Hányféleképpen tehetik ezt meg, ha

a) bármely elnökségi összetétel lehetséges?

a) a közkedvelt Kovács urat mindenképpen szeretnék beválasztani a vezetségbe?

b) a kiállhatatlan Kovács úrnak semmiképpen sem szeretnének tisztséget adni?

c) Kovács úr és felesége, Kovácsné nem kaphatnak egyszerre tisztséget (de külön-külön igen) ?

Megoldás

```
> restart; with(combinat):
```

a)

```
> n := 25;
  k := 3;
```

$$n := 25$$

$$k := 3 \qquad (2.3.10.1.1)$$

Egy $n=25$ tagú klubból kell $k=3$ tagot kiválasztani, és egy tag nem kaphat egyszerre több tisztséget (mert akkor nem lenne 3 tagú a vezetség). A sorrend számít, mert nem mindegy, hogy valaki elnök, titkár vagy jegyz lesz-e (pl. az 1. kiválasztott lesz az elnök, a 2. kiválasztott lesz a titkár, a 3. pedig a jegyz). Ez tehát egy ismétlés nélküli variáció, melynek kiszámítására az $\frac{n!}{(n-k)!}$ képlet alkalmazható. Így az elnökség lehetséges összetételeinek száma:

```
> nA := n!/(n-k)!;
```

$$nA := 13800 \qquad (2.3.10.1.2)$$

Ugyanerre az eredményre jutunk a beépített *numbperm* függvénnyel:

```
> numbperm(n,k);
```

$$13800 \qquad (2.3.10.1.3)$$

b)

Az a) feladatrész logikáját követhetjük egy kis módosítással. Vegyük észre, hogy az ismétlés nélküli variációk száma megkapható úgy is, hogy elbb sorrend nélkül kiválasztunk 3 tagot a 25-bl (ism. nélküli kombináció, $\binom{25}{3}$), majd sorba rendezzük ket (ism. nélküli permutáció, $3!$). Azaz $\frac{n!}{(n-k)!} = \binom{n}{k} \cdot k!$.

```
> nA = binomial(n, k)*k!;
```

$$13800 = 13800 \qquad (2.3.10.1.4)$$

Most, mivel Kovács úr mindenképpen kap tisztséget, már csak 2 elnökségi tagot kell választani mellé a maradék 24-b1 ($\binom{24}{2}$), majd a (Kovács úrral együtt) 3 választottat kell ismét sorbarendezni ($3!$). Így a variációk száma:

```
> nB := binomial(n - 1, k - 1)*k!;
```

$$nB := 1656 \qquad (2.3.10.1.5)$$

c)

A b) feladatrészhez hasonlóan okoskodhatunk. Kovács úr nem kaphat tisztséget, így a megmaradó 24 tagból kell 3-at kiválaszatni ($\binom{24}{3}$), majd sorbarendezni ($3!$):

```
> nC := binomial(n - 1,k)*k!;
```

$$nC := 12144 \qquad (2.3.10.1.6)$$

d)

Itt érdemes elbb az ellentétesemény eseteit megszámolni, majd kivonni az összes esetek számából.

Ellentétesemény: Kovács úr és Kovácsné is tisztséget kapnak. Ekkor már csak a 1 tagot kell kiválasztani a 23-ból, hogy teljes legyen az elnökség, majd a 3 tagot még sorba kell rendezni (ki milyen tisztséget kap):

```
> nDc := binomial(n - 2, k - 2)*k!; # = 23*3!
```

$$nDc := 138 \qquad (2.3.10.1.7)$$

A kedvez esetek száma az a) részben kapott összes esetek számának és az elbb kiszámolt kedveztlen esetek számának különbsége:

```
> nD := nA - nDc;
```

$$nD := 13662 \qquad (2.3.10.1.8)$$

Gyakorló feladatok

Gy1.

a) Hány 6 jegy szám alkotható az 1, 1, 2, 2, 2, 3 számjegyekbl?
b) Mekkora a valószínsége, hogy egy ilyen szám páros?

Gy2.

a) Hány különböz háromjegy szám alkotható az 1, 2, 3, 4, 5 számjegyekbl, ha egy számjegy többször is felhasználható?
b) Mekkora valószínséggel lesz a szám páros?

Gy3.

a) Hány osztója van a következ számoknak: 10, 11, 12, 13?
b) Általában hogyan számítható ki egy természetes szám osztóinak száma a prímtényezs felbontása ismeretében?

Gy4.

a) Egy házibulin egy 9 fs társaságból hányféleképp lehet összeállni egy 4 fs társasjátékhoz?
b) Eközben hányféleképp alakulhat ki a teraszon egy 5 fs beszélgetés?

Gy5.

a) Hányféleképp lyukasztható egy (9 cellás) buszjegy pontosan két lyukkal?
b) És pontosan három lyukkal?
c) És tetszleges számú lyukkal?

Gy6.

a) Hányféle eredménye lehet a klasszikus ötös lottó sorsolásának?
b) Mennyi esélyünk van egy kitöltött szelvénnyel a fnyereményre?

Gy7.

Hányféleképpen ülheti körbe 7 lovag a kerekasztalt, ha a forgatással egymásba viheto ülésrendeket azonosnak tekintjük?

Gy8.

Egy kis lakás 5 helyiségét akarjuk kifesteni (egyszínre), háromféle festékünk van. Hányféle színváltozat jöhet létre a teljes lakásra nézve?

Gy9.

Ha az ötös lottó sorsolását úgy végeznénk, hogy a húzott számokat mindig visszatesszük a következ húzás eltt, mekkora lenne a kockázata, hogy nem sikerül a végén öt különböz nyerszámot produkálni?

Gy10.

Hányféleképp olvasható ki egy négyzet alakú táblázatból a KOMBINATORIKA szó (az 5. feladatban látott elrendezés szerint)?